Actividad 8: Computo Simbolico con Maxima

Valenzuela Carrillo María Inés

31 de Marzo del 2016

1. Introducción

El sistema de álgebra computacional Maxima es un motor de cálculo simbólico escrito en lenguaje Lisp publicado bajo licencia GNU GPL.

Cuenta con un amplio conjunto de funciones para hacer manipulación simbólica de polinomios, matrices, funciones racionales, integración, derivación, manejo de gráficos en 2D y 3D, manejo de números de coma flotante muy grandes, expansión en series de potencias y de Fourier, entre otras funcionalidades.

Maxima es un descendiente de Macsyma, el sistema de álgebra computacional desarrollado a finales de 1960 en el Instituto Tecnológico de Massachusetts (MIT). Macsyma fue revolucionario en sus días y muchos sistemas posteriores, tales como Maple y Mathematica, se inspiraron en él.

En esta actividad se pide repetir uno de los ejercicios de cada una de las secciones del manual de Jay Kerns[3].

2. Geometría Tridimensional

2.1. Vectores y Álgebra Lineal

En esta sección se muestran diversas operaciones con vectores,como sumas, multiplicaciones por escalares, producto cruz y producto punto.

Ejemplo 1: utilizando el producto punto.

```
(%i1)  a: [4,7,9];
```

(%o1) $[4, 7, 9]$

```
(%i2)  b: [12,3,6];
```

(%o2) $[12, 3, 6]$

```
(%i3)  c:[1,8,7];
```

(%o3) $[1, 8, 7]$

```
(%i4)  (a.b)/(b.c)*a;
```

(%o4) $[\frac{82}{13}, \frac{287}{26}, \frac{369}{26}]$

2.2. Líneas, Planos y Superficies Cuadráticas

Se pueden graficar diversas superficies con Maxima utilizando el paquete draw.

Ejemplo 2: graficando un hiperboloide.

```
(%i1)  hiperboloide: x^2 + y^2 - z^2 = 1;
```

(%o1) $-z^2 + y^2 + x^2 = 1$

```
(%i2)  load(draw);
```

(%o2) $/usr/share/maxima/5.32.1/share/draw/draw.lisp$

```
(%i6)  draw3d(enhanced3d = true, implicit(hiperboloide, x,-2,2,
       y,-2,2, z,-1.5,1.5),palette=[6,7,8]);
```

(%o6) $[\mathrm{gr3d}\,(implicit)]$



2.3. Funciones Vectoriales

Se pueden definir funciones vectoriales utilizando Maxima, de manera muy similar a otras funciones.

Ejemplo 3: Dereivando funciones vectoriales.

```
(%i1) r(t) := [t^3, 3*cos(t), 4*sin(t)];
```

(%o1) $\mathrm{r}(t) := [t^3, 3\cos(t), 4\sin(t)]$

```
(%i2) diff(r(t), t);
```

(%o2) $[3\,t^2, -3\sin(t), 4\cos(t)]$

```
(%i3) define(rp(t), diff(r(t), t));
```

(%o3) $\mathrm{rp}(t) := [3\,t^2, -3\sin(t), 4\cos(t)]$

```
(%i4) float(rp(2));
```

(%o4) $[12.0, -2.727892280477045, -1.664587346188569]$

2.4. Longitud de arco y curvatura

No hay funciones especiales en Maxima para curvaturas, pero se pueden calcular con las formulas conocidad.

Ejemplo 4: Encontrando la curvatura κ de una función

```
(%i1) r(t) := [t, cos(t), sin(t)];
```

(%o1) $\mathrm{r}(t) := [t, \cos(t), \sin(t)]$

```
(%i2) rp(t) := [1 ,-sin(t), cos(t)];
```

(%o2) $\mathrm{rp}(t) := [1, -\sin(t), \cos(t)]$

```
(%i3) Tp(t) := [0, -cos(t), sin(t)]/sqrt(2);
```

(%o3) $\mathrm{Tp}(t) := \frac{[0, -\cos(t), \sin(t)]}{\sqrt{2}}$

```
(%i4) sqrt(Tp(t) . Tp(t))/sqrt(rp(t) . rp(t));
```

(%o4) $\frac{\sqrt{\frac{\sin(t)^2}{2} + \frac{\cos(t)^2}{2}}}{\sqrt{\sin(t)^2 + \cos(t)^2 + 1}}$

```
(%i5) trigsimp(%);
```

(%o5) $\frac{1}{2}$

```
(%i6) define(kappa(t), %);
```

(%o6) $\kappa(t) := \frac{1}{2}$

3. Funciones de Varias Variables

Maxima puede trabajar y graficar funciones de varias variables con sus curvas de nivel.

Ejemplo 5: Graficando funciones de varias variables.

```
(%i1)  f(x,y) := (3*x^2 + 4*y^2)^2;
```

(%o1) $f(x,y) := \left(3\,x^2 + 4\,y^2\right)^2$

```
(%i2)  load(draw);
```

(%o2) $/usr/share/maxima/5.32.1/share/draw/draw.lisp$

```
(%i3)  draw3d(palette= [7,8,9],enhanced3d = true, explicit(f(x,y), x,
       -3, 3, y, -3,3));
```

(%o3) [gr3d ($explicit$)]

La gráfica resultante es:



3.1. Derivadas Parciales

Las derivadas parciales se manejan de manera muy similar a las derivadas descritas anteriormente.

Ejemplo 6: Derivando parcialmente una función F, primero dos veces respecto a x y después la misma función F respecto a y.

```
(%i1) F: (4-x^2)/(7-x+y^2);
```

(%o1) $\frac{4-x^2}{y^2-x+7}$

```
(%i2) diff(F,x,2);
```

(%o2) $-\frac{2}{y^2-x+7}-\frac{4\,x}{(y^2-x+7)^2}+\frac{2\,(4-x^2)}{(y^2-x+7)^3}$

```
(%i3) diff(F,y);
```

(%o3) $-\frac{2\,(4-x^2)\,y}{(y^2-x+7)^2}$

3.2. Aproximación Lineal y Diferenciales

El plano tangente a una función en un punto es útil para encontrar valores de la función en la cercanía del punto, En Maxima de puede encontrar el plano tangente mediante la aproximación lineal utilizando la función de taylor.

Ejemplo 7: Encontrando la aproximación lineal para una función.

```
(%i1) f(x,y) := log(x^2)*cos(y)/exp(x^2);
```

(%o1) $\mathrm{f}\,(x,y) := \frac{\log(x^2)\,\cos(y)}{\exp(x^2)}$

```
(%i2) taylor(f(x,y), [x,y], [1,2], 2);
```

(%o2) $2\cos(2)\,(x-1)\,e^{-1}-\left(5\cos(2)\,(x-1)^2+2\sin(2)\,(y-2)\,(x-1)\right)\,e^{-1}+$ ...

3.3. Regla de la Cadena y Diferenciación Implicita

Cuando se tiene una función de varias variables, donde dichas variables son también funciones de variabes distintas de puede uilizar la regla de la cadena.

Ejemplo 8: Utilizando la regla de la cadena.

```
(%i7)  f(x,y) := log(x^2) * cos(y);
```

(%o7) $\mathrm{f}(x, y) := \log\left(x^2\right) \cos(y)$

```
(%i8)  [x,y] : [s * t^2, s * t^2];
```

(%o8) $[s\,t^2, s\,t^2]$

```
(%i9)  diff(f(x,y), s);
```

(%o9) $\dfrac{2\cos\left(s\,t^2\right)}{s} - t^2 \sin\left(s\,t^2\right) \log\left(s^2\,t^4\right)$

```
(%i10) diff(f(x,y), t);
```

(%o10) $\dfrac{4\cos\left(s\,t^2\right)}{t} - 2\,s\,t \sin\left(s\,t^2\right) \log\left(s^2\,t^4\right)$

```
(%i11) kill(x, y);
```

(%o11) *done*

```
(%i12) diff(f(x,y), x);
```

(%o12) $\dfrac{2\cos(y)}{x}$

```
(%i13) diff(f(x,y), y);
```

(%o13) $-2\log(x)\sin(y)$

3.4. Derivadas Direccionales y el Gradiente

Mediante Maxima se puede encontrar el gradiente de una función de manera sencilla.

Ejemplo 9: Encontrando el gradiente una función.

```
(%i1) f(x,y) := log(x^2) * cos(y);
```

(%o1) $\mathrm{f}(x,y) := \log\left(x^2\right)\,\cos(y)$

```
(%i2) load(vect);
```

(%o2) $/usr/share/maxima/5.32.1/share/vector/vect.mac$

```
(%i3) scalefactors([x,y]);
```

(%o3) $done$

```
(%i4) gdf: grad(f(x,y));
```

(%o4) $2\,\mathrm{grad}\left(\log(x)\,\cos(y)\right)$

```
(%i5) ev(express(gdf), diff);
```

(%o5) $[\frac{2\cos(y)}{x}, -2\log(x)\,\sin(y)]$

```
(%i6) define(gdf(x,y), %);
```

(%o6) $\mathrm{gdf}(x,y) := [\frac{2\cos(y)}{x}, -2\log(x)\,\sin(y)]$

3.5. Optimización y Extremos Locales

Para encontrar los puntos críticos se necesita resolver el sistema de ecuaciones $f_x = 0$ y $f_y = 0$.

Ejemplo 10: Encontrando los puntos críticos de una función.

```
(%i56) f(x,y) := 2 * (x-5)^2 + 2 * (y+4)^2 + x*y;
```

(%o56) $\mathrm{f}(x,y) := 2\,(x-5)^2 + 2\,(y+4)^2 + x\,y$

```
(%i57) fx : diff(f(x,y), x);
```

(%o57) $y + 4\,(x-5)$

```
(%i58) fy : diff(f(x,y), y);
```

(%o58) $4\,(y+4)+x$

```
(%i59) solve([fx,fy], [x,y]);
```

(%o59) $[[x=\frac{32}{5},y=-\frac{28}{5}]]$

```
(%i60) H: hessian(f(x,y), [x,y]);
```

(%o60) $\begin{pmatrix}4 & 1\\ 1 & 4\end{pmatrix}$

```
(%i61) determinant(H);
```

(%o61) 15

```
(%i62) subst([x = 32/5, y = -28/5], diff(fx, x));
```

(%o62) 4

```
(%i63) subst([x = 32/5, y = -28/5], determinant(H));
```

(%o63) 15

```
(%i64) f(32/5,-28/5);
```

(%o64) $-\frac{134}{5}$

3.6. Multiplicadores de Lagrange

Ejemplo 11: Usando multiplicadores de Lagrange.

```
(%i77) f(x,y) :=  (x-5)^2 + (y+4)^2;
```

(%o77) $\mathrm{f}\left(x,y\right):=\left(x-5\right)^2+\left(y+4\right)^2$

```
(%i78) g: x^2 + y^2;
```

(%o78) y^2+x^2

```
(%i79) eq1: diff(f(x,y), x) = h * diff(g, x);
```

(%o79) $2\,(x-5)=2\,h\,x$

```
(%i80) eq2: diff(f(x,y), y) = h * diff(g, y);
```

(%o80) $2\,(y+4)=2\,h\,y$

```
(%i81) eq3: g = 1;
```

(%o81) $y^2+x^2=1$

```
(%i82) solve([eq1, eq2, eq3], [x, y, h]);
```

(%o82) $[[x=-\frac{5}{\sqrt{41}},y=\frac{4}{\sqrt{41}},h=\sqrt{41}+1],[x=\frac{5}{\sqrt{41}},y=-\frac{4}{\sqrt{41}},h=1-\sqrt{41}]]$

```
(%i83) [f(-5/sqrt(41),4/sqrt(41)), f(5/sqrt(41),4/sqrt(41))];
```

(%o83) $[\left(\frac{4}{\sqrt{41}}+4\right)^2+\left(-\frac{5}{\sqrt{41}}-5\right)^2,\left(\frac{5}{\sqrt{41}}-5\right)^2+\left(\frac{4}{\sqrt{41}}+4\right)^2]$

4. Integración Múltiple

4.1. Integrales Dobles

Mediante Maxima es sencillo realizar integrales dobles.

Ejemplo 12: Realizando una integral doble definida.

```
(%i1) f(x,y) := cos(y)^3 - (x^2)*y;
```

(%o1) $\mathrm{f}(x,y):=\cos(y)^3-x^2\,y$

```
(%i2) integrate(integrate(f(x,y), y), x);
```

(%o2) $-\frac{x\sin(y)^3}{3}+x\sin(y)-\frac{x^3\,y^2}{6}$

```
(%i3) integrate(integrate(f(x,y), y, x^1/2, 2 - x), x, 0, 1);
```

(%o3) $-\frac{40\cos(2)^3+240\cos(2)-40\cos(1)^3-240\cos(1)-80\cos\left(\frac{1}{2}\right)^3-480\cos\left(\frac{1}{2}\right)+647}{360}$

4.2. Integración en Coordenadas Polares

Se puede integrar n coordenadas polares simplemente haciendo el cambio.

Ejemplo 13: Integrando en coordnadas polares.

```
(%i1) f(x,y) := x^3 + y^3;
```

(%o1) $\mathrm{f}(x,y) := x^3+y^3$

```
(%i2) [x,y]: [r * cos(theta), r * sin(theta)];
```

(%o2) $[r\cos(\theta), r\sin(\theta)]$

```
(%i3) integrate(integrate(f(x,y) * r, r, 0, 2*cos(theta)), theta,-%pi/2,%pi/2);
```

(%o3) $\frac{7\,\pi}{4}$

4.3. Integrales Triples

Las integrales triples se realizan de manera similar a las dobles.

Ejemplo 14: Realizando una integral triple definida.

```
(%i1) f(x,y,z) := x^2+y^2+z^2;
```

(%o1) $\mathrm{f}(x,y,z) := x^2+y^2+z^2$

```
(%i2) integrate(integrate(integrate(f(x,y,z),z,0,x+y),y,0,-x),x,0,1);
```

(%o2) $\frac{-2}{15}$

4.4. Integrales en coordenadas Esféricas y Cilíndricas

De igual manera que en las coordenas polares solo se realiza el cambio.

Ejemplo 15: Integrando en coordenadas cilíndricas.

```
(%i1) f(x,y,z) := x^2 + y^2 +z;
```

(%o1) $\mathrm{f}(x, y, z) := x^2 + y^2 + z$

```
(%i2) [x,y,z] : [r*cos(theta), r*sin(theta), z];
```

(%o2) $[r\cos(\theta), r\sin(\theta), z]$

```
(%i3) integrate(integrate(integrate(f(x,y,z)*r, z,0,3), r,0,2), theta,0,%pi);
```

(%o3) $\frac{90\pi + 256}{10}$

4.5. Cambio de Variable

El cambio de variable es una herramienta muy útil para facilitar la integración, Maxima permite hacerlo de manera sencilla.

Ejemplo 16: Integrando con cambio de variable.

```
(%i11) f(x,y) := x^2+y^2;
```

(%o11) $\mathrm{f}(x, y) := x^2 + y^2$

```
(%i12) [x,y]: [u - v^4, 5 * u * v];
```

(%o12) $[u - v^4, 5\,u\,v]$

```
(%i13) J: jacobian([x,y], [u,v]);
```

(%o13) $\begin{pmatrix} 1 & -4\,v^3 \\ 5\,v & 5\,u \end{pmatrix}$

```
(%i14) J: determinant(J);
```

(%o14) $20\,v^4 + 5\,u$

```
(%i15) integrate(integrate(f(x,y) * J, u,1,2), v,3,4);
```

(%o15) $\dfrac{2645574283}{26}$

5. Calculo Vetorial

5.1. Campos Vectoriales

Un campo vectorial es una función vectorial que asigna a cada punto un vector.

Ejemplo 17: Graficando un campo vectorial.

```
(%i1)  load(draw);
```

(%o1) $/usr/share/maxima/5.32.1/share/draw/draw.lisp$

```
(%i12) coord: setify(makelist(k,k,-4,4));
```

(%o12) $-4, -3, -2, -1, 0, 1, 2, 3, 4$

```
(%i13) points2d: listify(cartesian_product (coord ,coord ));
```

```
(%i14) vf2d(x,y):=  vector ([x,y],[x,y]);
```

(%o14) $\mathrm{vf2d}(x, y) := \mathrm{vector}([x, y], [x, y])$

```
(%i15) vect2: makelist(vf2d(k[1],k[2]),k,points2d );
(%i16) apply(draw2d , append ([ head_length =0.1, color=red], vect2 ));
```

La gráfica resultante es:



Figura 1: Campo Vectorial

5.2. Integrales de Linea

Ejemplo 18: Realizando una integral de linea.

```
(%i1)  F(x,y,z) := [2*x*y*z, (x^2)*z, (x^2)*y ];
```

(%o1) $\mathrm{F}\left(x,y,z\right) := [2\,x\,y\,z, x^2\,z, x^2\,y]$

```
(%i2)  [x,y,z]: [t^2, t+4, 1/t];
```

(%o2) $[t^2, t+4, \frac{1}{t}]$

```
(%i3)  romberg(F(x,y,z) . diff([x,y,z], t), t, 0, 1);
```

(%o3) 5.0

5.3. Campos Vectoriales Conservativos

Se puede saber si un campo es conservativo utilizando la función curl, cuando el rotacional es 0 el campo es conservativo.

Ejemplo 19: Encontrando el rotacional de un no campo conservativo y uno conservativo respectivamente.

(%i1) `F(x,y) := [4*x^3 - 5*y^2, 5*y^3 - 3*x];`

(%o1) $F(x,y) := [4\,x^3 - 5\,y^2, 5\,y^3 - 3\,x]$

(%i2) `load(vect);`

(%o2) $/usr/share/maxima/5.32.1/share/vector/vect.mac$

(%i3) `scalefactors([x,y]);`

(%o3) $done$

(%i4) `curl(F(x,y));`

(%o4) $\mathrm{curl}\left([4\,x^3 - 5\,y^2, 5\,y^3 - 3\,x]\right)$

(%i5) `express(%);`

(%o5) $\frac{d}{d\,x}\left(5\,y^3 - 3\,x\right) - \frac{d}{d\,y}\left(4\,x^3 - 5\,y^2\right)$

(%i6) `ev(%, diff);`

(%o6) $10\,y - 3$

(%i7) `F(x,y) := [x^3 + 5*y, 5*y^3 + 5*x];`

(%o7) $F(x,y) := [x^3 + 5\,y, 5\,y^3 + 5\,x]$

(%i8) `ev(express(curl(F(x,y))), diff);`

(%o8) 0

Referencias

[1] Wikipedia, the free encyclopedia, *Maxima (software)*.

[2] Actividad 8 (2016-1), `http://computacional1.pbworks.com/w/page/106192917/Actividad8(2016-1)`.

[3] Multivariable Calculus with Maxima, *G. Jay Kerns*, `http://gkerns.people.ysu.edu/maxima/maximaintro/maximaintro.pdf`.